NAG Library Routine Document

F08JAF (DSTEV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08JAF (DSTEV) computes all the eigenvalues and, optionally, all the eigenvectors of a real symmetric n by n tridiagonal matrix A.

2 Specification

```
SUBROUTINE F08JAF(JOBZ, N, D, E, Z, LDZ, WORK, INFO)
INTEGER           N, LDZ, INFO
double precision  D(*), E(*), Z(LDZ,*), WORK(*)
CHARACTER*1       JOBZ
```

The routine may be called by its LAPACK name ***dstev***.

3 Description

F08JAF (DSTEV) computes all the eigenvalues and, optionally, all the eigenvectors of A using a combination of the QR and QL algorithms, with an implicit shift.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphiahttp://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOBZ – CHARACTER*1 *Input*

On entry: if JOBZ = 'N', compute eigenvalues only.

If JOBZ = 'V', compute eigenvalues and eigenvectors.

Constraint: JOBZ = 'N' or 'V'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix.

Constraint: N ≥ 0.

3: D$(*)$ – ***double precision*** array *Input/Output*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: the n diagonal elements of the tridiagonal matrix A.

On exit: if INFO = 0, the eigenvalues in ascending order.

4: E$(*)$ – ***double precision*** array *Input/Output*

Note: the dimension of the array E must be at least $\max(1, \mathrm{N}-1)$.

On entry: the $(n-1)$ subdiagonal elements of the tridiagonal matrix A.

On exit: the contents of E are destroyed.

5: Z(LDZ,$*$) – ***double precision*** array *Output*

Note: the second dimension of the array Z must be at least $\max(1, \mathrm{N})$.

On exit: if JOBZ = 'V', then if INFO = 0, Z contains the orthonormal eigenvectors of the matrix A, with the ith column of Z holding the eigenvector associated with D(i).

If JOBZ = 'N', Z is not referenced.

6: LDZ – INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08JAF (DSTEV) is called.

Constraints:

if JOBZ = 'V', LDZ $\geq \max(1, \mathrm{N})$;
otherwise LDZ ≥ 1.

7: WORK$(*)$ – ***double precision*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, 2 \times \mathrm{N}-2)$.

On exit: if JOBZ = 'N', WORK is not referenced.

8: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, the algorithm failed to converge; i off-diagonal elements of E did not converge to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(A+E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the ***machine precision***. See Section 4.7 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating-point operations is proportional to n^2 if JOBZ = 'N' and is proportional to n^3 if JOBZ = 'V'.

9 Example

This example finds all the eigenvalues and eigenvectors of the symmetric tridiagonal matrix

$$A = \begin{pmatrix} 1 & 1 & 0 & 0 \\ 1 & 4 & 2 & 0 \\ 0 & 2 & 9 & 3 \\ 0 & 0 & 3 & 16 \end{pmatrix},$$

together with approximate error bounds for the computed eigenvalues and eigenvectors.

9.1 Program Text

```
*     F08JAF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=20)
      INTEGER          LDZ
      PARAMETER        (LDZ=NMAX)
*     .. Local Scalars ..
      DOUBLE PRECISION EERRBD, EPS
      INTEGER          I, IFAIL, INFO, N
*     .. Local Arrays ..
      DOUBLE PRECISION D(NMAX), E(NMAX-1), RCONDZ(NMAX), WORK(2*NMAX-2),
     +                 Z(LDZ,NMAX), ZERRBD(NMAX)
*     .. External Functions ..
      DOUBLE PRECISION X02AJF
      EXTERNAL         X02AJF
*     .. External Subroutines ..
      EXTERNAL         DDISNA, DSTEV, X04CAF
*     .. Intrinsic Functions ..
      INTRINSIC        ABS, MAX
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08JAF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the diagonal and off-diagonal elements of the matrix A
*        from data file
*
         READ (NIN,*) (D(I),I=1,N)
         READ (NIN,*) (E(I),I=1,N-1)
*
*        Solve the symmetric tridiagonal eigenvalue problem
*
         CALL DSTEV('Vectors',N,D,E,Z,LDZ,WORK,INFO)
*
         IF (INFO.EQ.0) THEN
*
*           Print solution
*
            WRITE (NOUT,*) 'Eigenvalues'
            WRITE (NOUT,99999) (D(I),I=1,N)
*
            IFAIL = 0
            CALL X04CAF('General',' ',N,N,Z,LDZ,'Eigenvectors',IFAIL)
*
*           Get the machine precision, EPS and compute the approximate
*           error bound for the computed eigenvalues.  Note that for
*           the 2-norm, max( abs(D(i)) ) = norm(A), and since the
*           eigenvalues are returned in ascending order
*           max( abs(D(i)) ) = max( abs(D(1)), abs(D(n)))
*
```

```
            EPS = X02AJF()
            EERRBD = EPS*MAX(ABS(D(1)),ABS(D(N)))
*
*           Call DDISNA (F08FLF) to estimate reciprocal condition
*           numbers for the eigenvectors
*
            CALL DDISNA('Eigenvectors',N,N,D,RCONDZ,INFO)
*
*           Compute the error estimates for the eigenvectors
*
            DO 20 I = 1, N
               ZERRBD(I) = EERRBD/RCONDZ(I)
   20       CONTINUE
*
*           Print the approximate error bounds for the eigenvalues
*           and vectors
*
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Error estimate for the eigenvalues'
            WRITE (NOUT,99998) EERRBD
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Error estimates for the eigenvectors'
            WRITE (NOUT,99998) (ZERRBD(I),I=1,N)
         ELSE
            WRITE (NOUT,99997) 'Failure in DSTEV. INFO =', INFO
         END IF
      ELSE
         WRITE (NOUT,*) 'NMAX too small'
      END IF
*
99999 FORMAT (3X,(8F8.4))
99998 FORMAT (4X,1P,6E11.1)
99997 FORMAT (1X,A,I4)
      END
```

9.2 Program Data

```
F08JAF Example Program Data

  4                          :Value of N

  1.0  4.0  9.0  16.0  :End of diagonal elements
  1.0  2.0  3.0        :End of off-diagonal elements
```

9.3 Program Results

```
 F08JAF Example Program Results

 Eigenvalues
     0.6476  3.5470  8.6578 17.1477
 Eigenvectors
           1       2       3       4
 1   0.9396  0.3388  0.0494  0.0034
 2  -0.3311  0.8628  0.3781  0.0545
 3   0.0853 -0.3648  0.8558  0.3568
 4  -0.0167  0.0879 -0.3497  0.9326

 Error estimate for the eigenvalues
        1.9E-15

 Error estimates for the eigenvectors
        6.6E-16    6.6E-16    3.7E-16    2.2E-16
```